The Resouce Manager Controller and Transaction Management in COMCodebook

Introduction

The Resource Manager Controller (RMC) in COMCodebook provides the key service of managing MTS Transactions that cross multiple components. This document will discuss the purpose and implementation of the RMC and its place in a distributed system. We will then continue with a description of the different classes used in a VFP front-end. And we’ll finish with a walk-through that describes what occurs during a save.

The Resource Manager Controller

When a particular function such as Save occurs under MTS it often requires that the save process actually save multiple resources. For instance, saving a new invoice may involve changes to a customer , invoice, invoice detail and inventory resources. In order to accomplish this cross component trasactioning in MTS we must have one component marked as Requires New Transaction and subservient transactions that are marked as either Supports Transactions or Requires Transations. These subservient components enroll in the main transaction. The RMC is the component that manages the main transaction. When a save is issued by the user interface it actually packages all of the necessary resources and sends them to the RMC which begins a transaction then delegates the actual processing of that transaction to the various resource managers.

VFP Front End Development using COM Codebook

One of the design goals for COMCodebook was to enable the developer to create interfaces that are as complex and functional as the ones they created in the previous version of Codebook. This means that a developer has to be able to define any number of parent child relationships in a variety of different configurations. In this manner, datasources and resource managers can be defined once, with the relationships between resources defined at run-time by an application. This means that once a set of middle-tier components have been created for a company, adding a new front-end simply consists of defining the relations between those components and providing an appropriate UI.

Addressing this requirement for a VFP front-end is the domain of the Resource Manager Proxy (RMP), the Resource Manager Proxy Loader (RMPL) and the Resource Manager Proxy Controller (RMPC). 

The Resource Manager Proxy

The RMP is an Interface level representation of the Resource Manager component (that exists in the middle tier. Its purpose is to talk to the middle tier component and to translate any resources from the middle tier into a front-end appropriate format. For instance, it may translate a disconnected ADO Recordset into a VFP Cursor so that a VFP front-end developer can work in a manner that they are used to.

The RMP also allows stores parent and child relationships, messages to display if problems occur, information about parent primary key and children foreign key fields, etc. Basically it is intended to function like a good portion of the old Codebook business object.

The Resource Manager Proxy Loader

The RMPL is simply a loader mechanism that allows a developer to visually drop RMP’s onto a form so that they will be instantiated at form run-time. Using the properties of the cResourceManagerProxyLoader, located in CUTILS.VCX, a developer can specify what RMP to load, and any parent/child relationships that should be created.

The Resource Manager Proxy Controller

The RMPC keeps track of all RMPs in a given form. As an RMP is created, it registers itself with the RMPC. After the last RMP is loaded, the RMPC takes on the responsibility of matching up children RMPs objects with their parents. The RMPC is the direct equivalent of the middle tier RMC and handles all of the communication with it.

Step by Step through a Form

Resource Form Initialization

When a form is instantiated, the Resource Manager Proxy Controller is created in the Load() of the form.

*---------------- Location Section ---------------------
*} Library: CFORMS.VCX
*} Class: CResourceForm
*} Method: Load()
*-------------------------------------------------------
*) Description: 
*) Creates the Resource Manager Conrol Object
*-------------------------------------------------------

DODEFAULT()
THIS.CreateResourceManagerControlObject()

The following code illustrates how the framework identifies which object the form should use for its Resource Manager Proxy Controller. It uses the name of the class definition stored in the .cResourceManagerControlClass property. The default value for this property is "cResourceManagerProxyCollection" which is stored in the CCOLLECT class library (it uses "iResourceManagerProxyCollection" in the I-Layer). The form stores the resultant object in its own .oRMControl property.

*---------------- Location Section ---------------------
*} Library: CFORMS.VCX
*} Class: cBaseForm
*} Method: CreateResourcemanagerControlObject()
*-------------------------------------------------------
*) Description:
*) Creates the object that tracks and controls communication
*) between the many Resource Mananger Proxys that are
*) present on the form.
*-------------------------------------------------------
WITH THIS
IF !EMPTY( .cResourceManagerControlClass )
.oRMControl = CREATEOBJECT( .cResourceManagerControlClass, THIS )
ENDIF
ENDWITH

RETURN .T.

Once the form loads it is now time for the objects contained on the form to load as well. Among these objects are the Resource Manager Proxy Loaders. The class definition for these objects is located in the CUTILS.VCX and is named CResourceManagerProxyLoader, feel free to jump over there and take a peak at the Init() method if you so desire. There, you will see that the primary function of the RMProxyLoader is to create the Resource Manager Proxy objects and register them with the form’s RMController. It does this by passing a reference of the newly created Resource Manager Proxy object to the controller’s .Add() method.

Adding a Resource Manager Proxy to the Form’s RMPC

The Resource Manager Proxy Controller object stores a reference to all of the Resource Manager Proxy objects contained on the form. The .Add() method is the primary means of registering the proxy objects. Four basic tasks are performed in this method. First, the controller is examined to determine if the RM Proxy was already added. Second, if it is not already a part of the collection a new property is created and a reference to the Proxy is stored therein. Third, a row is added to the .aChildren array of the controller and it stores the name of the newly added proxy. Finally, the control count is incremented by one. 

Linking Parent and Child Resource Manager Proxy objects

Once all of the Resource Manager Proxy objects are created and registered with the form, processing returns to the form’s Init() method where the form instructs the .oRMController that it is now time to introduce the children resources to their parents. 

*---------------- Location Section ---------------------
*} Library: CFORMS.VCX
*} Class: CResourceForm
*} Method: Init()
*-------------------------------------------------------
*) Description:
*-------------------------------------------------------

LOCAL llRetVal
llRetVal = DODEFAULT()
IF llRetVal
THIS.oRMControl.LinkParentChildResources()
ENDIF

RETURN llRetVal

In the LinkParentChildResource() method, the Resource Manager Proxy Collection scans each registered RMP and accomplishes two important tasks. First, it stores the name of the RMP’s corresponding middle tier Resource Manager in column 2 of the array. Second, it determines whether or not the current proxy object has a parent. If so, it registers the child with the parent by calling the parent’s .RegisterChildObject() method.

Note, that the RMPC now knows each RMP on the form, what relation it has to any other, and for what middle-tier component each RMP is a proxy.

Saving Information

Now that we have an RMPC with all of our required information, let’s see what happens when the user presses the Save button.

First, the form receives the Save message through its .Save() method and passes it directly to the Resource Manager Proxy Controller.

*---------------- Location Section ---------------------
*} Library: CFORMS.VCX
*} Class: CResourceForm
*} Method: Save()
*-------------------------------------------------------
*) Description:
*) This method is responsible for the following things:
*) a. Locking the form during the saving processing
*) b. Unlocking the form after the save processing
*) is complete.
*) c. Passing the save message through to the 
*) Resource Manager Proxy Controller.
*) d. Informing the application that a Save
*) type of activity took place.
*-------------------------------------------------------

LOCAL lnRetVal, llOldLockScreen
=LockScreen( .T., @llOldLockScreen )
lnRetVal = THIS.oRMControl.Save()
THISFORM.Refresh()
=LockScreen( llOldLockScreen )
goApp.RecordActivity( "FormSave" )

RETURN lnRetVal

Next, the Resource Manager Proxy Controller’s Save() method takes over and accomplishes the following: 

*---------------- Location Section ---------------------
*} Library: CCOLLECT.VCX
*} Class: CResourceManagerProxyCollection
*} Method: Save()
*-------------------------------------------------------
*) Description:
*) Template method that controls the saving the information
*) in the user interface to the data source on the 
*-------------------------------------------------------
LOCAL lnRetVal, loRMController, lcXML
lnRetVal = FILE_CANCEL
IF THIS.oPrimaryResource.AllowSave()
IF THIS.oPrimaryResource.WriteBuffer()
IF .NOT. THIS.oPrimaryResource.IsAdding()
IF .NOT. THIS.oPrimaryResource.IsChanged()
RETURN FILE_CANCEL
ENDIF
ENDIF

DO CASE
CASE .NOT. THIS.CreateRMController( @loRMController )
CASE .NOT. THIS.GetEnvDefinition( @lcXML )
CASE .NOT. THIS.SetUpEnvironment( loRMController, lcXML )
CASE .NOT. ( THIS.MarshallData( loRMController ) == FILE_OK )
OTHERWISE
THIS.oPrimaryResource.BeforeSave()
THIS.oPrimaryResource.oControlContainer.BeforeSave()
lnRetVal = loRMController.Save()
THIS.oPrimaryResource.oControlContainer.AfterSave( lnRetVal )
THIS.oPrimaryResource.oControlContainer.RefreshControls( THIS.oPrimaryResource )
THIS.oPrimaryResource.AfterSave( lnRetVal )
ENDCASE
ENDIF
ENDIF
IF lnRetVal = FILE_OK
goApp.DisplayMessage( THIS.oPrimaryResource.cSaveSuccessMessageKey )
ELSE
goApp.DisplayMessage( THIS.oPrimaryResource.cSaveFailureMessageKey )
ENDIF

RETURN lnRetVal

If you have never been introduced to one of these animals before, please, allow me. Reader, meet a Template Method … Template Method … meet Reader. A template method does one thing and one thing only. It defines WHAT is executed and in WHAT order. It does NOT define HOW anything is accomplished. The HOW is defined in other methods, which can be easily overridden if necessary. Furthermore, a relatively inexperienced reader should be able to figure out its intended functionality in a relatively short period of time just by reading the method calls. The benefit to the developer is that it allows you to modify processing by overriding a specific method in a subclass without having to override the template method itself. For example, if you don’t like how we CreateRMController() you can define your own processing for creating your RM Controller in the I Layer by overriding your .CreateRMController() method. 

In essence, this method asks each RMP whether or not things on the front-end are in shape to save. If this is true we get to the heart of the Save method.

The first step in the actual save process is to create a reference to the Resource Manager Controller object located on the middle tier server (.CreateRMController() method). Next, it gets a definition of the environment defined by the programmer in the interface. This is a list of all resource manager proxy objects defined in the environment, the name of their parent objects (if any) as well as which object is the primary resource manager for this configuration. Next, it tells the Resource Manager Controller on the middle-tier to recreate this environment using middle-tier Resource Manager objects instead of UI-based Resource Manager Proxy objects. Finally, it then marshals all of the data from the interface over to the corresponding objects on the middle tier. 

If all of these tasks completed successfully the Resource Manager Proxy Controller calls the .Save() method on to the Resource Manager Controller.

* CAbstractRMController in CResourceManagers.PRG 
*--------------
FUNCTION Save()
*--------------

LOCAL lnRetVal, loContext
lnRetVal = FILE_CANCEL

THIS.CreateMTSContextObject( @loContext )
WITH THIS.oPrimaryResource
IF .Validate()
lnRetVal = .Save()
ENDIF
ENDWITH

THIS.ExecuteMTSSetOperations( loContext, lnRetVal )
RETURN lnRetVal
ENDFUNC

Notice that this middle-tier method finally enables the MTS transaction by creating the MTS Context Object. It then tells the primary resource to validate itself. If the primary resource (and all of its children) pass the validation test then the Save() method executes. Once the Save() method executes the result is passed to the ExecuteMTSSetOperatons() method where the transaction is either aborted or committed. A Sequence diagram of the Save is listed below.

Sequence Diagram

Conclusion

The physical separation of the data by the individual tiers, the requirement to use MTS Transactions when updating data and the requirement that developers be able to create complex parent child relationships in the interface presented some interesting challenges. These challenges were met using a controller object on both the interface and business tiers. The developer defines the intricacies of their resource manager proxy object’s relationships on the interface layer. That environment is automatically recreated in the business layer, the data is transferred (marshaled) to the business layer where it is wrapped in an MTS transaction, validated and finally saved. 

